MetaPost

MetaPost refers to both a programming language and the interpreter of the MetaPost programming language. Both are derived from Donald Knuth's Metafont language and interpreter. MetaPost produces vector graphic diagrams from a geometric/algebraic description. The language shares Metafont's declarative syntax for manipulating lines, curves, points and geometric transformations. However,
Many of the limitations of MetaPost derive from features of Metafont. For instance, MetaPost does not support all features of PostScript. Most notably, paths can have only one segment (so that regions are simply connected), and regions can be filled only with uniform colours. PostScript level 1 supports tiled patterns and PostScript 3 supports Gouraud shading.
Availability and usage.
MetaPost is distributed with many distributions of the TeX and Metafont framework, for example, it is included in the MiKTeX and the TeX Live distributions.
The encapsulated postscript produced by Metapost can be included in LaTeX, ConTeXt, and TeX documents via standard graphics inclusion commands. The encapsulated postscript output can also be used with the PDFTeX engine, thus directly giving PDF. This ability is implemented in ConTeXt and in the LaTeX graphics package, and can be used from plain TeX via the supp-pdf.tex macro file.
ConTeXt and LuaTeX supports the inclusion of MetaPost code within the input file. Inclusion of MetaPost code in LaTeX is also possible by using LaTeX-packages, for example gmp or mpgraphics. 
Examples.
This is a single file <samp>example.mp</samp> which when processed by the MetaPost interpreter (via the command codice_1 on Linux) produces three eps files <samp>example.1</samp>, <samp>example.2</samp>, <samp>example.3</samp>. These are pictured on the right.
transform pagecoords;
pagecoords:=identity scaled 10mm shifted (100mm,150mm);
beginfig (1)
 fill ((0,0)--(2,0)--(2,1)--(1,1)--(1,2)--(0,2)--cycle)
 transformed pagecoords withcolor green;
 draw ((2,0)..(2,1)..(1,1)..(1,2)..(0,2))
 transformed pagecoords;
 drawarrow ((0,0)--(2,2)) transformed pagecoords;
endfig;
beginfig (2)
 draw (for i=0 upto 7: dir (135i)-- endfor cycle)
 transformed pagecoords;
endfig;
pagecoords:=identity scaled 15mm shifted (100mm,150mm);
beginfig (3);
 % declare paths to be used
 path p[],p[]t;
 % set up points by defining relationships
 z1=(0,0); z2=z1+2up;
 z3=z1+whatever*dir (60)=z2+whatever*dir (-50);
 z4=z3+(-1.5,-.5);
 z5=z1+dir (135);
 z0=whatever[z1,z2]=whatever[z3,z4];
 % set up paths
 p0=fullcircle yscaled .5 rotated 45 shifted z0 ;
 p1=z2---z4..z0..z3---z1;
 p2=p1 cutbefore p0 cutafter p0;
 p3=p0 cutbefore p1 cutafter p1;
 p4=p2---p3---cycle;
 % define transformed versions of paths and points
 for i=0 upto 4: p[i]t=p[i] transformed pagecoords; endfor
 for i=0 upto 5: z[i]t=z[i] transformed pagecoords; endfor
 % do some drawing
 fill p4t withcolor (1,1,0.2);
 draw z1t---z2t withcolor .5white;
 draw z3t---z4t withcolor .5white;
 pickup pencircle;
 draw p0t dashed withdots scaled .3;
 draw p1t dashed evenly;
 draw p2t withcolor blue;
 draw p3t withcolor red;
 label.lrt (btex $z_0$ etex, z0t);
 label.llft (btex $z_1$ etex, z1t);
 label.top (btex $z_2$ etex, z2t);
 label.rt (btex $z_3$ etex, z3t);
 label.llft (btex $z_4$ etex, z4t);
 for i=0 upto 4:
 drawdot z[i]t withpen pencircle scaled 2;
 endfor
endfig;
bye
The resulting three eps files can be used in TeX via LaTeX's codice_2 command, ConTeXt's codice_3, Plain TeX's codice_4 command, or (in Plain pdftex) the codice_5 command from <samp>supp-pdf.tex</samp>. To view or print the third diagram, this inclusion is necessary, as the TeX fonts (Computer Modern) are not included in the eps files produced by MetaPost by default.